INTEGRATED CIRCUITS

APPLICATION NOTE

ABSTRACT

Presents short and simple I²C software routines that support only slave (rather than master or master & slave) operation and an ASM demonstration program. The slave-only software in this app note complements the master mode software presented in AN464, *Using the 87LPC76X microcontroller as an I²C bus master*.

AN463 I²C slave routines for the 87LPC76X

Author: Bill Houghton

2000 Jan 11



AN463

Author: Bill Houghton

The 87LPC76X Microcontroller combines in a small package the benefits of a high-performance microcontroller with on-board hardware supporting the Inter-Integrated Circuit (I²C) bus interface.

The 87LPC76X can be programmed both as an I^2C bus master, a slave, or both. An overview of the I^2C bus and description of the bus support hardware in the 87LPC76X microcontrollers appears in application note AN464, *Using the 87LPC76X Microcontroller as an I*²C Bus Master. That application note includes a programming example, demonstrating a bus-master code. Here we show an example of programming the microcontroller as an I²C slave.

The code listing demonstrates communications routines for the 87LPC76X as a slave on the I²C bus. It compliments the program in AN464 which demonstrates the 87LPC76X as an I²C bus master. One may demonstrate two 87LPC76X devices communicating with each other on the I²C bus, using the AN464 code in one, and the program presented here in the other. The examples presented here and in AN464 allow the 87LPC76X to be either a master or a slave, but not both. Switching between master and slave roles in a multimaster environment is described in application note AN435.

The software for a slave on the bus is relatively simple, as the processor plays a relatively passive role. It does not initiate bus transfers on its own, but responds to a master initiating the communications. This is true whether the slave receives or transmits data—transmission takes place only as a response to a bus master's request. The slave does not have to worry about arbitration or about devices which do not acknowledge their address. As the slave is not supposed to take control of the bus, we do not demand it to resolve bus exceptions or "hangups". If the bus becomes inactive the processor simply withdraws, not interfering with the master (or masters) on the bus which should (hopefully) try to resolve the situation.

The 87LPC76X has a single bit l^2 C hardware interface where the registers may directly affect the levels on the bus, and the software interacting with the hardware registers takes part in the protocol implementation. The hardware and the low level routines dealing with the registers are tightly coupled. We repeat here the warning from the 87LPC76X bus-master application note: one should take extra care if trying to modify these lower level routines.

The service routine for the I²C slave is interrupt driven per message. This allows for master communication requests which are not synchronized with the application program running on the slave. It is possible to write simple slave application programs which will not be interrupt driven, taking care not to lose master transmissions while doing something else, but the user should be discouraged from doing so. As the slave should respond to asynchronous requests of masters on the bus, an interrupt driven service routine makes sense—and, as the code demonstrates, is simple to implement.

DEMONSTRATION CODE

The main program operation, intended for demonstration only, is simple. There are two data buffers, one for data reception and one for data transmission. When new data has been received from the I^2C bus into the receive buffer, the program writes it into the transmit buffer. The first byte of received data is copied to Port 1. When a bus master requests to read data, Port 0 will be returned for the first byte of requested data, while the remaining bytes will come from the transmit buffer. This allows for simple testing of a master and slave system by having the master compare data received to data sent. This scheme also allows the 87LPC76X to be used as a one-byte I^2C I/O port.

The program begins at address 0, where the microprocessor begins execution after a hardware reset. This location contains a jump instruction to the main program, which starts at the label Reset (towards the end of the listing). Upon reset, the program initializes the stack pointer, the I²C address of the slave processor (MyAddr) and clears the data buffers and software flags. In this program the receive and transmit buffers are each eight bytes long—the maximum number of bytes is defined by the label MaxBytes. One may easily change the program to handle longer messages by changing the value of MaxBytes and allocating more data memory to the buffers.

The I²C interface is configured to operate as a slave by setting the msb of register I²CFG. This is done simultaneously with loading the appropriate value of CTVAL—bits CT0 and CT1, which are determined by the frequency of the microprocessor's crystal. The interface hardware is explicitly instructed to get into the slave idle mode by setting the appropriate bit in the I2CON register. Timer I, which operates as a "watchdog" timer detecting bus hangups, is activated and its interrupts are enabled.

After the initialization, the program gets to the label MainLoop. Most of the time the program will "hang" in a wait loop at this label, simply waiting for an I²C interrupt to occur. When there is an I²C bus request there will be an interrupt, the service routine will be executed and we shall return to the MainLoop label. If the service routine receives new data, it sets a flag, DatFlag, signalling that data has been updated. This flag will allow us to leave the MainLoop label, and execute a short routine copying the updated input buffer to the output (transmit) buffer.

If a new bus interrupt comes before overwriting of the old read buffer data is completed, and an undesirable "mix" of old and new data might occur. This type of situation is avoided by disabling the I^2C interrupts (clearing the IE2 bit in the Interrupt Enable 1 Register) just before copying the data to the transmit buffer, and re-enabling the interrupts when the copy operation is completed.

When the copy routine is completed the DatFlag is cleared and we jump back to MainLoop, waiting for the next interrupt to occur. If the interrupt is for data transmission the service routine will not set DatFlag, and upon return we shall remain at the MainLoop label.

THE INTERRUPT SERVICE ROUTINE

The service routine is interrupt driven with respect to the start of each I^2C frame, but within each frame the interaction with the hardware is based on polling. An occurrence of a Start on the bus will cause an interrupt that will initiate the service routine which starts at address 33H. After saving registers, all interrupts except the I^2C interrupt itself are enabled, as we want to allow response to other interrupts during the routine. The philosophy behind this is that the I^2C may be a lower priority than some other operations in the system. Since the I^2C hardware will stretch the clock until the program responds, an interrupt of reasonable duration will not have a harmful effect on the data transfer.

Since we intend to react to the l^2C hardware by polling the ATN flag in wait loops, we do not want the expected changes on the bus to take us again to the beginning of the routine. Therefore, the El2 flag is cleared, masking further l^2C interrupts even when interrupts are re-enabled (by the ACALL to a RETI instruction).

At the label Slave, the routine starts receiving the address on the bus. Each new address bit is read after a software wait loop detects that the ATN flag is set by the hardware. Note that with the single bit implementation of the I²C port, the software must closely support the hardware: for example, we need to explicitly clear the Start status before we enter a wait loop for the next bit. If the software does not clear the Start flag, the hardware will stretch the low period of the clock (SCL line) on the bus—and the first address bit will simply not occur. (Such a state will not go on forever—eventually the processor will release the bus as a result of a Timer I timeout.)

Reception of the eight bits of Address + R/W is completed using part of the receive byte subroutines. The address received is compared to MyAddr, the address of this specific slave. If the address is different the processor goes idle and leaves the service routine. If the message is intended for this processor (received address matches MyAddr) the Read/Write bit is tested, and the program jumps to the appropriate labels. When the R/W bit is low the master requests a Write—and this slave should receive the data written into it. When the R/W bit is high the master is requesting a Read and this slave should transmit the data (at code label Read).

For "Master Write" we send an acknowledge for the address byte and proceed with receiving the data bytes, responding with an acknowledge for each and transferring them into the receive buffer. For long messages, when the buffer is full (we have received MaxByte bytes) we read from the bus one additional byte and then send a negative acknowledge, letting the master know it should stop sending us data. Then we set DatFlag to signal the mainline program that new data has been received, and jump to MsgEnd. At the MsgEnd label we wait for the next Stop or Repeated Start. On a Stop we resume the idle mode (Goldle) and return from the service routine. On a Restart the slave process starts again with reception of the new address at the label Slave.

If the message is short enough so that the receive buffer is not filled up, the RcvByte subroutine (called after WrtLoop) will return due to the Stop condition, DRDY will not be set, and we shall exit the loop via label WLEx—setting the DatFlag and proceeding to MsgEnd.

For "Master Read" the transmit buffer is sent on the bus byte by byte in the RdLoop, using the XmitByte subroutine. We exit the loop when all the buffer is transmitted, or the Master does not respond with an acknowledge. Note that lack of acknowledgement for slave transmission does not necessarily indicate a problem or that the receiving master is busy. This could very well be a normal operation of the protocol, which defines that a receiving master signals the transmitting slave to end its message by explicitly transmitting a negative acknowledge as a response to the last byte the master is interested in. The protocol does not include inherent means for specifying in advance the length of a requested message.

SUBROUTINES

The lower level subroutines closely interact with the hardware and the activity on the bus. The XmitByte subroutine transmits one byte and receives the acknowledge bit that comes in response. The byte receive routine, which one may use from different entry points, receives a data or an address byte, and takes care of acknowledgements. When a Start or Stop is detected the subroutine returns immediately—the calling routine is expected to check the flags to determine whether a whole byte has been received (DRDY will be set), or a Start or a Stop condition has occurred.

Close inspection of RcvByte code shows that a total of nine bits are being read off the bus. The first bit does not belong to the received byte, but is the acknowledge this processor sent in response of the former byte or address. Reading the Ack bit from the I2DAT register clears the Transmit Active state and DRDY, thus releasing SCL and allowing the bus activity to proceed to the next data bit. Upon return the Ack bit is left in the Carry flag, and the actual data byte received is returned in the Acc register.

Upon Timer I interrupt code execution commences at address 73H, where there is a jump to the service routine Timerl. This interrupt is caused by the watchdog timer, as a result of an I^2C bus that is "hanging" without activity in the middle of a transmission for too long a period of time. The slave simply clears the bus interface, and starts all over again at the label Reset.

I²C slave routines for the 87LPC76X

AN463

*****	* * * * * * * *	* * * * * * * * * * * * * * * * * * * *	******
;			s for the 87LPC764
; This pr		monstrates I2C slave	functions for the 87LCP764. It is a for the 8xC751/752 in AN430.
; eight b	ytes dee	p. The sample main p	and receive data buffers that are each rogram copies received data to the ed data can be read back by a bus master.
; the por ; port 0 ; transmi ; data by	t output pin data ssion, a tes may	s, except for the I2 . The code will acce dditional bytes will	1, such that an I2C write will affect C pins P1.6 & P1.7. An I2C read will return pt only eight data bytes in any one I2C not be acknowledged. Similarly, only eight 2C transmission. This program does not cess.
;*******	* * * * * * * *	* * * * * * * * * * * * * * * * * * * *	******
; – I ; – T ; – I ; – I	2C inter imer I i ENO SFR 2C inter	-	
\$mod764 \$debug			
;******	* * * * * * * *	* * * * * * * * * * * * * * * * * * * *	*******
; Value d	efinitic	ns.	
CTVAL	equ	02h	; CT1, CT0 bit values for I2C.
MaxBytes	equ	8	; Max # of bytes to be sent or recvd.
SlvAdr ;SlvAdr	equ equ	7Eh 76h	; 7Eh is the keypad on I2C demo bd. ; 76h is the LED display on I2C demo bd.
; Masks f	or I2CFG	bits.	
BTIR BSLAV	equ equ	10h 80h	; Mask for TIRUN bit. ; Mask for Slave Enable bit.
; Masks f	or I2CON	bits.	
BCXA BIDLE BCDR BCARL BCSTR BCSTP	equ equ equ equ equ	80h 40h 20h 10h 08h 04h	<pre>; Mask for CXA bit. ; Mask for IDLE bit. ; Mask for CDR bit. ; Mask for CARL bit. ; Mask for CSTR bit. ; Mask for CSTP bit.</pre>
; RAM loc	ations u	sed by I2C routines.	
RcvDat	data	10h	; I2C receive data buffer (8 bytes). ; addresses 10h through 17h.
XmtDat	data	18h	; I2C transmit data buffer (8 bytes). ; addresses 18h through 1Fh.
Flags	data	20h	; I2C software status flags.
NoAck	bit	Flags.7	; Holds negative acknowledge flag.
DatFlag	bit	Flags.6	; Tells whether an I2C write operation ; has occurred.

AN463

```
BitCnt
         data
               21h
                                ; I2C bit counter.
ByteCnt
         data
                22h
                                 ; Send/receive byte counter.
MyAddr
         data
                24h
                                 ; Holds address of THIS slave.
AdrRcvd
         data
                25h
                                ; Holds received slave address + R/W.
               AdrRcvd.0
                                ; Slave read/write flag.
RWFlaq
         bit
Begin Code
******
; Reset and interrupt vectors.
                                 ; Reset vector at address 0.
         ajmp
               Reset
; I2C interrupt is used to detect a start while the slave is idle.
                                 ; I2C interrupt.
         org
                33h
         push
                psw
                                 ; Save status.
                                ; Save accumulator.
         push
               acc
         clr
                ei2
                                ; Disable I2C interrupt.
              ClrInt
                                ; Re-enable other interrupts.
         acall
               Slave
         ajmp
; Timer I timeout interrupt service routine.
                0073h
         orq
                                ; Timer I interrupt address.
                                ; Clear timer I interrupt.
               CLRTI
TimerI:
         setb
                I2CFC,#0
                                 ; Turn off I2C.
         mov
                                              ; Reset I2C flags.
         mov
                I2CON, #BCXA+BCDR+BCARL+BCSTR+BCSTP
         clr
               TIRUN
         acall ClrInt
                                ; Clear interrupt pending.
                                ; Re-start.
         ajmp
               Reset
ClrInt:
         reti
         orq
                0100h
Main Transmit and Receive Routines
Slave:
               I2CON,#BCARL+BCSTP+BCSTR+BCXA ; Clear start status.
         mov
         jnb
                ATN,$
                                ; Wait for next data bit.
         mov
               BitCnt,#7
                                ; Set bit count.
         acall
               RcvB2
                                ; Get remainder of slave address.
               AdrRcvd,A
         mov
                                ; Save received address + R/W bit.
         clr
               acc.0
         cjne
               A,MyAddr,GoIdle
                                ; Enter idle mode if not our address.
         jb
                RWFlag,Read
                                ; Read or Write?
         mov
               R0,#RcvDat
                                ; Set up receive buffer pointer.
                ByteCnt, #MaxBytes ; Max 4 bytes can be received.
         mov
WrtLoop:
         acall
                SendAck
                                ; Send acknowledge.
         acall
                RcvByte
                                 ; Get data byte from master.
                DRDY,WLEx
                                ; Must be end of frame?
         inb
         mov
                @R0,A
                                ; Save data.
         inc
                R0
                                ; Advance buffer pointer.
         djnz
                ByteCnt,WrtLoop ; Back to receive if buffer not full.
         acall
                SendAck
                                ; Send acknowledge.
         acall RcvByte
                                ; Get, but do not store add'l data.
               I2DAT,#80h
                                ; Send negative acknowledge.
         mov
         jnb
                ATN,$
                                ; Wait for acknowledge sent.
WLEX:
               DatFlag
                                ; Flag main that data has been received.
         setb
                                 ; Buffer full, enter idle mode.
         sjmp
               MsgEnd
```

AN463

Read:	mov mov acall	R0,#XmtDat ByteCnt,#MaxBytes SendAck	; Set up transmit buffer pointer.; Max bytes to be sent.; Send address acknowledge.
RdLoop:	mov cjne mov	A,@R0 R0,#XmtDat,RdL1 A,P0	; Get data byte from buffer. ; Return port 1 value instead of buffer ; data if this is buffer address 0.
RdL1:	inc acall jb djnz MOV	R0, XmitByte NoAck,RLEx ByteCnt,RdLoop A,P3	 Advance buffer pointer. Send data byte. Exit if NAK. Back if more data requested & avail. data if this is buffer address 1.
RLEx:	sjmp	MsgEnd	; Done, enter idle mode.
MsgEnd:	jnb JB	ATN,\$ STR,Slave	<pre>; Wait for stop or repeated start. ; If repeated start, go to slave mode, ; else enter idle mode.</pre>
GoIdle:	mov pop pop setb ret	I2CON, #BCSTP+BCXA+BC ACC PSW EI2	CDR+BCARL+BIDLE ; Enter slave idle mode. ; Restore accumulator. ; Restore status. ; Re-enable I2C interrupts.
;*******	* * * * * * * * *	* * * * * * * * * * * * * * * * * * * *	******
; ;*******	* * * * * * * * *	Subrout ******	lines
; Byte tran ; Enter w		utine. a in ACC.	
XmitByte: XmBit:	mov mov rl jnb djnz mov jnb mov ret	I2DAT,A A ATN,\$ BitCnt,XmBit	<pre>; Set 8 bits of data count. ; Send this bit. ; Get next bit. ; Wait for bit sent. ; Repeat until all bits sent. ; Switch to receive mode. ; Wait for acknowledge bit. ; Save acknowledge bit.</pre>
; RcvByte ; RcvB2 ;	k : sends e : rece : rece 7]	s an I2C acknowledge. ives a byte of data.	of I2C data, used to allow reception of
SendAck:	mov jnb ret	12DAT,#0 ATN,\$	<pre>; Send receive acknowledge. ; Wait for acknowledge sent.</pre>
RcvByte: RcvB2: RBit: RBEx:	mov clr orl rl jnb jnb djnz mov rlc ret	BitCnt,#8 A A,I2DAT A ATN,\$ DRDY,RBEx BitCnt,RBit C,RDAT A	<pre>; Set bit count. ; Init received byte to 0. ; Get bit, clear ATN. ; Shift data. ; Wait for next bit. ; Exit if not a data bit. ; Repeat until 7 bits are in. ; Get last bit, don't clear ATN. ; Form full data byte.</pre>

AN463

;		Mair	n Program
;*******	* * * * * * * *		***************************************
Reset:	mov	SP,#2Fh	; Set stack location.
ICEDCC -	mov	R0,#RcvDat	; Set up pointer to data area.
	mov		
D		R1,#2*MaxBytes	; Set up buffer length counter.
RLoop:	mov	@R0,#0	; Clear buffer memory.
	inc	R0	; Advance to next buffer position.
	djnz	R1,RLoop	; Repeat until done.
	mov	AdrRcvd,#0	
	mov	MyAddr,#SlvAdr	; Set our slave address.
	mov	Flags,#0	; Clear system flags.
	setb	EI2	; Enable I2C interrupt.
	setb	ETI	; Enable Timer I interrupt.
	setb	EA	; Enable interrupt system.
	mov	I2CFG, #BSLAV+CTVA	
	mov		<pre>IP+BXCA+BCDR+BCARL_BIDLE ; Put slave into idle mod</pre>
	setb	TIRUN	; Turn on timer I.
: the in	nout buf	fer to the output h	operation. It Also copies the rest of ouffer at the same time, acting like a
; small	memory	device.	buffer at the same time, acting like a
; small MainLoop:	memory jnb	device. DatFlag,\$	ouffer at the same time, acting like a ; Wait for data sent from I2C.
	memory	device.	buffer at the same time, acting like a
; small MainLoop:	memory jnb	device. DatFlag,\$	ouffer at the same time, acting like a ; Wait for data sent from I2C.
; small MainLoop:	memory jnb mov	device. DatFlag,\$ pcon,#01h EA A,RcvDat+1	<pre>ouffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location)</pre>
; small MainLoop:	memory jnb mov clr	device. DatFlag,\$ pcon,#01h EA	ouffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move.
; small MainLoop:	memory jnb mov clr mov	device. DatFlag,\$ pcon,#01h EA A,RcvDat+1	<pre>ouffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location)</pre>
; small MainLoop:	memory jnb mov clr mov orl	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch</pre>	<pre>puffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster.</pre>
; small MainLoop:	memory jnb mov clr mov orl mov	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A</pre>	<pre>puffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1.</pre>
; small MainLoop:	memory jnb mov clr mov orl mov mov	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat</pre>	<pre>buffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1. ; Set input buffer start pointer. ; Set output buffer start pointer.</pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes</pre>	<pre>buffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1. ; Set input buffer start pointer. ; Set output buffer start pointer. ; Set buffer length counter.</pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov orl mov mov mov mov mov	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0</pre>	<pre>buffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1. ; Set input buffer start pointer. ; Set output buffer start pointer. ; Set buffer length counter. ; Get data from input buffer.</pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov mov mov mov mov	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A</pre>	<pre>buffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1. ; Set input buffer start pointer. ; Set output buffer start pointer. ; Set buffer length counter. ; Get data from input buffer. ; Store data in output buffer.</pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov mov mov mov mov inc	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A R1</pre>	<pre>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>></pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov mov mov mov mov inc inc	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A R1 R0</pre>	<pre>buffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1. ; Set input buffer start pointer. ; Set output buffer start pointer. ; Set buffer length counter. ; Get data from input buffer. ; Store data in output buffer. ; Increment input buffer pointer. ; Increment output buffer pointer.</pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov mov mov mov mov inc inc djnz	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A R1 R0 R2,ML2</pre>	<pre>buffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1. ; Set input buffer start pointer. ; Set output buffer start pointer. ; Set buffer length counter. ; Get data from input buffer. ; Store data in output buffer. ; Increment input buffer pointer. ; Repeat until entire buffer is updated.</pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov mov mov mov mov inc inc	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A R1 R0</pre>	<pre>buffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1. ; Set input buffer start pointer. ; Set output buffer start pointer. ; Set buffer length counter. ; Get data from input buffer. ; Store data in output buffer. ; Increment input buffer pointer. ; Increment output buffer pointer.</pre>
; small MainLoop:	memory jnb mov clr mov orl mov mov mov mov mov mov mov mov inc inc djnz	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A R1 R0 R2,ML2</pre>	<pre>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>></pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov mov mov mov inc inc djnz clr	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A R1 R0 R2,ML2 DatFlag</pre>	<pre>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>></pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov mov mov mov inc inc djnz clr setb	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A R1 R0 R2,ML2 DatFlag EA</pre>	<pre>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>></pre>
; small MainLoop: ; ***	memory jnb mov clr mov orl mov mov mov mov mov mov mov mov setb sjmp	<pre>device. DatFlag,\$ pcon,#01h EA A,RcvDat+1 a,#0Ch P1,A R0,#RcvDat R1,#XmtDat R2,#MaxBytes A,@R0 @R1,A R1 R0 R2,ML2 DatFlag EA MainLoop</pre>	<pre>buffer at the same time, acting like a ; Wait for data sent from I2C. ; Enter Idle Mode. ; Turn off interrupts during data move. ; Get first data byte (second buffer location) ; Mask off I2C pins to prevent disaster. ; Store data to port 1. ; Set input buffer start pointer. ; Set output buffer start pointer. ; Set buffer length counter. ; Get data from input buffer. ; Store data in output buffer. ; Increment input buffer pointer. ; Increment output buffer pointer. ; Repeat until entire buffer is updated. ; Clear I2C transmission flag. ; Wait for next I2C transmission.</pre>

end

AN463

Definitions

Short-form specification — The data in a short-form specification is extracted from a full data sheet with the same type number and title. For detailed information see the relevant data sheet or data handbook.

Limiting values definition - Limiting values given are in accordance with the Absolute Maximum Rating System (IEC 134). Stress above one or more of the limiting values may cause permanent damage to the device. These are stress ratings only and operation of the device at these or at any other conditions above those given in the Characteristics sections of the specification is not implied. Exposure to limiting values for extended periods may affect device reliability.

Application information - Applications that are described herein for any of these products are for illustrative purposes only. Philips Semiconductors make no representation or warranty that such applications will be suitable for the specified use without further testing or modification.

Disclaimers

Life support — These products are not designed for use in life support appliances, devices or systems where malfunction of these products can reasonably be expected to result in personal injury. Philips Semiconductors customers using or selling these products for use in such applications do so at their own risk and agree to fully indemnify Philips Semiconductors for any damages resulting from such application.

Right to make changes — Philips Semiconductors reserves the right to make changes, without notice, in the products, including circuits, standard cells, and/or software, described or contained herein in order to improve design and/or performance. Philips Semiconductors assumes no responsibility or liability for the use of any of these products, conveys no license or title under any patent, copyright, or mask work right to these products, and makes no representations or warranties that these products are free from patent, copyright, or mask work right infringement, unless otherwise specified.

Philips Semiconductors 811 East Arques Avenue P.O. Box 3409 Sunnyvale, California 94088-3409 Telephone 800-234-7381

© Copyright Philips Electronics North America Corporation 2000 All rights reserved. Printed in U.S.A.

Date of release: 01-00

Document order number:

9397 750 06848

Let's make things better.



